UNIVERSITÀ DELLA CALABRIA
DIPARTIMENTO DI MATEMATICA E INFORMATICA

Nuclear Medicine Scheduling via Answer Set Programming

Carmine Dodaro[1], Giuseppe Galatà[2], Cinzia Marte[1], Marco Maratea[1], and Marco Mochi[3]

[1] UNICAL, UNIVERSITY OF CALABRIA, RENDE, ITALY
[2] SURGIQ SRL, GENOVA, ITALY
[3] DIBRIS, UNIVERSITY OF GENOVA, GENOVA, ITALY

CILC 2024
Roma – June 28, 2024

Nuclear Medicine



Nuclear medicine is a branch of radiology used to:

- diagnose and determine a variety of diseases via small amounts of radioactive material
- study many types of cancers, heart disease, gastrointestinal, endocrine, neurological disorders, etc.

Nuclear Medicine

How does it work?

- The radioactive materials are introduced into the body by injection, swallowing, or inhalation.
- Once in the body, these substances accumulate in the target organs or tissues.
- Special cameras, such as PET or SPECT, detect the radiation emitted by these substances to create detailed images of the inside of the body.

Nuclear Medicine

Challenges in Treating Patients with Radiopharmaceuticals

- It is a complex process involving multiple resources and requiring multiple steps
  - Patients may need to follow specific instructions before and after procedures to ensure accurate results and minimize radiation exposure
- The radioactive elements are characterized by short half-lives
  - They decay rapidly after their preparation

Nuclear Medicine

Challenges in Treating Patients with Radiopharmaceuticals

- The scheduling needs to be as precise as possible to ensure high-quality images
- An efficient scheduling can:
  - increase the use of resources, avoiding waste of time
  - reduce the waiting time of the patients
  - increase the satisfaction of the patients

Nuclear Medicine

Proposed Solution: Efficient Scheduling with ASP

- We provide an efficient schedule for patient's treatments
  - To increase the level of care offered
- We propose a solution exploiting Answer Set Programming (ASP)
  - To easily model problem’s specifications

Nuclear Medicine Scheduling Problem

Nuclear Medicine Scheduling Problem

Problem Definition

- The Nuclear Medicine Scheduling (NMS) problem involves assigning patients to a specific day for their
  - medical check
  - preparation
  - the actual image detection process
- The schedule of the patients considers
  - the different requirements of the patients
  - varying time required for different procedures
  - number of injection chairs and tomographs available

Nuclear Medicine Scheduling Problem

Problem Description

- Each day is divided into 10 hours, 120 time slots of 5 minutes each
- Two rooms, each with:
  - one tomograph
  - three injection chairs
- 11 different protocols
- Each patient needs an exam and each exam is linked to a protocol defining the phases and the time required for each phase
- Each exam comprises four phases:

  (p1) anamnesis

  (p2) medical check

  (p3) radiopharmaceuticals injection and bio-distribution time

  (p4) image detection

Nuclear Medicine Scheduling Problem

Problem Description

Goal:

- Assign a starting and ending time to each scheduled phase

Requirements:

- There must be at most two patients concurrently in the medical check phase
- The injection phase must be done in an injection chair or on a tomograph according to the required protocol
- Each injection chair and tomograph can be used by just one patient at the same time
- Patients requiring an injection chair must be assigned to the tomograph of the same room
- Protocol identified by the id 815 cannot be assigned on the same day and tomograph for more than one patient

Nuclear Medicine Scheduling Problem

Problem Description

Optimizations:

- Maximize the number of scheduled patients in the considered days
- Minimize the unnecessary time spent in the clinic by the patients

NMS Problem Encoding

Nuclear Medicine Scheduling Problem

Encoding

The ASP program takes in input atoms representing:

- A list of patients, each requiring a specific protocol, to be assigned in a day
- The features of each exam
- Number of available rooms, tomographs, and injection chiars

The ASP program produces in output atoms representing:

- The assignments for each patient registration for the exam, detailing the phase and scheduled time slot for the day
- The allocated resource (either a chair or a tomograph) assigned to the patient during the specific day

Nuclear Medicine Scheduling Problem

Encoding

Assignment of a registration to the each phase

```
0 {x(ID, D, TS, PrID, 0) : avail(TS, D)} 1 :- reg(ID, D, PrID).
{x(ID, D, START, PrID, P+1) : avail(START,D), START >= TS+NumTS, START <
    TS+NumTS+6} = 1 :- x(ID, D, TS, PrID, P), exam(PrID, P, NumTS), P >= 0,
    P < 3.
:- x(ID, _, TS, PrID, 3), exam(PrID, 3, NumTS), TS + NumTS > 120.
```

There must be at most two patients concurrently in the medical check phase

```
timeAnamnesis(ID, TS..TS+NumTS-1) :- x(ID, D, TS, PrID, 0), exam(PrID, 0,
    NumTS).
:- #count{ID: timeAnamnesis(ID, TS)} > 2, avail(TS,D).
```

Nuclear Medicine Scheduling Problem

Encoding

Resource allocation: at most one patient is assigned to each tomograph and chair in every time slot

```
timeOccupation(ID, D, TS, END-1, PrID) :- x(ID, D, TS, PrID, 1), x(ID, D,
    END, PrID, 3).
res(ID, D, TS..END,0) :- timeOccupation(ID, D, TS, END, PrID),
    required_chair(PrID).
res(ID, D, TS..TS+NumTS-1,1) :- x(ID, D, TS, PrID, 3), exam(PrID, 3,
    NumTS), required_chair(PrID).
res(ID, D, TS..END+NumTS-1,1) :- timeOccupation(ID, D, TS, END, PrID),
    exam(PrID, 3, NumTS), not required_chair(PrID).
chair(C, ID, D, TS) :- chair(C, ID, D), res(ID, D, TS, 0).
tomograph(T, ID, D, TS) :- tomograph(T, ID, D), res(ID, D, TS, 1).
:- #count{ID: tomograph(T, ID, D, TS)} > 1, tomograph(T,_), avail(TS,D).
:- #count{ID : chair(C, ID, D, TS)} > 1, chair(C,_), avail(TS,D).
```

Nuclear Medicine Scheduling Problem

Encoding

Patients requiring an injection chair must be assigned to the tomograph of the same room

```
1 {chair(C, ID, D) : chair(C, _)} 1 :- x(ID, D, _, PrID, _),
    required_chair(PrID).
1 {tomograph(T, ID, D) : tomograph(T, _)} 1 :- x(ID, D, _, PrID, _).
:- chair(C, ID, D), tomograph(T, ID, D), chair(C, R1), tomograph(T, R2), R1
    != R2.
```

Checks the limit of protocols executed on a single tomograph

```
:- #count{ID: tomograph(T, ID, D), x(ID, D, _, PrID, _)} > N, limit(PrID,
    N), tomograph(T,_).
```

Nuclear Medicine Scheduling Problem

Encoding

Optimizations

```
:~ not x(ID, D, _, _,0), reg(ID, D, _). [1@2, ID, D]
:~ x(ID, _, START, PrID, 0), x(ID, _, END, _, 3), cost(PrID, NumTS), END -
    START - NumTS >= 0. [END - START - NumTS@1, ID]
```

Testing and Preliminary Results

Nuclear Medicine Scheduling Problem

Testing

The data used are real data coming from a medium size hospital provided by Medipass.

Instances Tested:

- Over a year of daily exams (366 instances)
- Excluding weekends, resulting in 72 weeks of data

Nuclear Medicine Scheduling Problem

Testing

Exams:

- 85% of patients require one specific protocol:
  - 2 time slots for anamnesis
  - 10 time slots for drug injection and bio-distribution
  - 2 time slots for medical preparation
  - 7 time slots for image detection
- Other patients are assigned to one of 10 possible protocols

Patient Statistics:

- Average of 29 patients scheduled per day
- Maximum of 37 patients scheduled in a day

Nuclear Medicine Scheduling Problem

Preliminary Results



- The majority of solutions have a patient assignment rate exceeding 80%
- Some solutions exhibit a low percentage of assigned patients due to constraints imposed by the data
  - E.g. restrictions of protocol with id 815

Nuclear Medicine Scheduling Problem

Preliminary Results



Analysis of waiting time based on the number of patients:

- Up to 29 patients: All instances solved optimally within the time limit.
- 36 and 37 patients: Encoding not able to optimally schedule within the time limit.
- Overall: Optimal solution found for more than 60% of instances

Future Works

- Extend the program taking into account
  - Fairness of the scheduling
  - Priority among patients
- Implement a web application
- Provide the solution of the Nuclear Medicine Rescheduling Problem

Thank you for the attention!